
1 Introduction

There is a pleasing simplicity of the computation as proof search paradigm
when based on inference rules. But how do we now implement it? One
way would be to capture the rules as propositions so we can represent a
program as a collection of propositions, which become the antecedents of
a sequent. The query will then be the succedent, and we can hopefully use
our usual tools of logical reasoning the obtain an operational semantics.

But what is the connection between rules and propositions, and can
we reformulate the bottom-up search strategy using inference rules instead
with antecedent propositions? This is the subject of today’s lecture.

2 Rules as Propositions

Let’s review a set of rules for increment in binary (omitting some fine points
on standard representations of numbers).

inc(e, b1(e))
ince

inc(b0(M), b1(M))
inc0

inc(M,N)

inc(b1(M), b0(N))
inc1

The first one ince is trival to express as just an atomic proposition; the
second (inc0) requires a simple quantifier. The third one has a premise, in
which case we can rewrite the inference rule as an implication. Let’s write

the resulting collection of antecedents as Γinc.

Γinc = inc(e, b1(e)),
∀m. inc(b0(m), b1(m)),
∀m.∀n. inc(m,n)⊃ inc(b1(m), b0(n))

We don’t yet have the tools to see in which way this translation is correct,
but at least in an intuitive sense it should look plausible.

Now, solving a logic programming query such as inc(b1(e), b1(b0(e)))
will be modeled as

Γinc
f−→ inc(b1(e), b1(b0(e)))

where “
f−→” demands a sequent calculus we have yet to design. Here we

use f to suggest focusing, although in this lecture we will just introduce
chaining. Focusing, eventually, will turn out to be inversion + chaining. The
particular form of chaining from this lecture is called backward chaining; in
the next lecture we also discuss forward chaining.

3 Chaining

We notice first that all propositions in Γinc are noninvertible: neither atoms,
nor universal quantification, nor implication have invertible rules as an-
tecedents. So the inversion strategy will not be useful. The second thing
we notice is that if we still had rules, we would know immediately that only
the inc1, represented as the third antecedent, will actually help in the exam-
ple at the end of the previous section. That is, we need to “look beyond”
the quantifiers and to the right of the implication until we find an atomic
formula and then decide if it would match our goal.

To formalize this, we focus on a particular proposition among the an-
tecedents and then continue to apply rules only to the proposition in focus
until we can determine whether it matches the succedent. We write [A] for
the particular proposition A that is in focus. There can be at most one of
those in a sequent.

The first rule picks a proposition from Γ (which, in our case, would be
Γinc) and puts it into focus. For now, the succedent will always be an atom,
so we wrote P .

A ∈ Γ Γ, [A]
f−→ P

Γ
f−→ P

focusL

The second rule instantiates a universal quantifier. Of course, at this point
we cannot know which term we might want to choose, so we instantiate it
with a variable X , to be determined later.

Γ, [A(X)]
f−→ P

Γ, [∀x.A(x)]
f−→ P

∀L

When we reach an implication B ⊃A, we would like to “look past” it at A,
but we also have to remember that we still have to prove B in case this A
does match the succedent.

Γ, [A]
f−→ P Γ

f−→ [B]

Γ, [B ⊃A]
f−→ P

⊃L

Two remarks: (1) we put Γ, [A]
f−→ P as the first premise (reversing the

order we have used so far for implication) because we would like to know
if it matches the succedent before we solve B, and (2) the focus is inherited
by both subformulas A and B.

With this process, we may finally come upon an atom Q, in which case
it has to unify with the succedent P .

Q = P

Γ, [Q]
f−→ P

id

Here, we imagine that Q = P is a unification which instantiates the free
variables like X we have introduced into the proof. Note that the substitu-
tion for such variables has to be global throughout the partial proof tree. If
we wanted to be more formal about this (although I don’t believe it really
helps understanding), we would “thread through” a substitution or a set
of constraints through all the rules.

The following observation is critical: if Q and P do not unify, then no rule
applies here.

no rule if Q 6= P

Γ, [Q]
f−→ P

Along the way we have introduced a second judgment form, Γ
f−→ [A],

where the focus on the right-hand side. For the example so far, we only

need one rule, where we lose focus.

Γ
f−→ P

Γ
f−→ [P ]

blur

Here is a summary of the rules so far:

A ∈ Γ Γ, [A]
f−→ P

Γ
f−→ P

focusL

Γ, [A(X)]
f−→ P

Γ, [∀x.A(x)]
f−→ P

∀L
Γ, [A]

f−→ P Γ
f−→ [B]

Γ, [B ⊃A]
f−→ P

⊃L

Q = P

Γ, [Q]
f−→ P

id
no rule if Q 6= P

Γ, [Q]
f−→ P

Γ
f−→ P

Γ
f−→ [P ]

blur

4 1+1=2

Now we apply the rules to check the result incrementing 1 in binary form.
Recall the encoding of the rules

Γinc = inc(e, b1(e)),
∀m. inc(b0(m), b1(m)),
∀m.∀n. inc(m,n)⊃ inc(b1(m), b0(n))

and the goal sequent, as yet without proof
...

Γinc
f−→ inc(b1(e), b0(b1(e)))

First we try inc(e, b1(e)):
...

Γinc, [inc(e, b1(e))]
f−→ inc(b1(e), b0(b1(e)))

Γinc
f−→ inc(b1(e), b0(b1(e)))

focusL

We fail immediately, because we are in the situation of the form Γ, [Q]
f−→

P where Q 6= P .
Next we try the second proposition in Γinc:

...

Γinc, [∀m. inc(b0(m), b1(m))]
f−→ inc(b1(e), b0(b1(e)))

Γinc
f−→ inc(b1(e), b0(b1(e)))

focusL

At this point we use a variable M for m, postpone a choice until some unifi-
cation constraints help us determine a good instantiation of the quantifier.

...

Γinc, [inc(b0(M), b1(M))]
f−→ inc(b1(e), b0(b1(e)))

Γinc, [∀m. inc(b0(m), b1(m))]
f−→ inc(b1(e), b0(b1(e)))

∀L

Γinc
f−→ inc(b1(e), b0(b1(e)))

focusL

Again we fail, this time because b0(M) does not unify with b1(e). Third
time’s a charm:

...

Γinc, [∀m.∀n. inc(m,n)⊃ inc(b1(m), b0(n))]
f−→ inc(b1(e), b0(b1(e)))

Γinc
f−→ inc(b1(e), b0(b1(e)))

focusL

Combining two consecutive ∀L rules, we arrive at

...

Γinc, [inc(M,N)⊃ inc(b1(M), b0(N))]
f−→ inc(b1(e), b0(b1(e)))

Γinc, [∀m.∀n. inc(m,n)⊃ inc(b1(m), b0(n))]
f−→ inc(b1(e), b0(b1(e)))

∀L× 2

Γinc
f−→ inc(b1(e), b0(b1(e)))

focusL

Now we are forced into the ⊃L rule. The incomplete proof state is now

...

Γinc, [inc(b1(M), b0(N))]
f−→ inc(b1(e), b0(b1(e)))

...

Γinc
f−→ [inc(M,N)]

Γinc, [inc(M,N)⊃ inc(b1(M), b0(N))]
f−→ inc(b1(e), b0(b1(e)))

⊃L

Γinc, [∀m.∀n. inc(m,n)⊃ inc(b1(m), b0(n))]
f−→ inc(b1(e), b0(b1(e)))

∀L× 2

Γinc
f−→ inc(b1(e), b0(b1(e)))

focusL

At this point, the first branch succeeds with unification, with M = e and
N = b1(e).

inc(b1(M), b0(N)) = inc(b1(e), b0(b1(e)))

Γinc, inc(b1(M), b0(N))
f−→ inc(b1(e), b0(b1(e)))

id

...

Γinc
f−→ [inc(M,N)]

Γinc, [inc(M,N)⊃ inc(b1(M), b0(N))]
f−→ inc(b1(e), b0(b1(e)))

⊃L

Γinc, [∀m.∀n. inc(m,n)⊃ inc(b1(m), b0(n))]
f−→ inc(b1(e), b0(b1(e)))

∀L× 2

Γinc
f−→ inc(b1(e), b0(b1(e)))

focusL

This substitution has to be applied globally to the partial proof, which
yields

Γinc, inc(b1(e), b0(b1(e)))
f−→ inc(b1(e), b0(b1(e)))

id

...

Γinc
f−→ [inc(e, b1(e))]

Γinc, [inc(e, b1(e))⊃ inc(b1(e), b0(b1(e)))]
f−→ inc(b1(e), b0(b1(e)))

⊃L

Γinc, [∀m.∀n. inc(m,n)⊃ inc(b1(m), b0(n))]
f−→ inc(b1(e), b0(b1(e)))

∀L× 2

Γinc
f−→ inc(b1(e), b0(b1(e)))

focusL

Singling out the remaining subgoal, we can solve this now by focusing on
the first of the propositions in Γinc. Trying to focus on any other one of the

antecedents will fail.

Γinc, [inc(e, b1(e))]
f−→ inc(e, b1(e))

id

Γinc
f−→ inc(e, b1(e))

focusL

Γinc
f−→ [inc(e, b1(e))]

blurR

Notice that in the example we could have added many many other
propositions in the antecedent and it would not have changed the outcome
as long as the eventual proposition at the end of the chain of quantifiers and
implications is not of the form inc( , ). In an implementation of a logic pro-
gramming language such as Prolog we compile the programs (that is, the
antecedents) so that we have direct access to those defining the particular
predicate in the query (that is, the succedent).

5 Horn Clauses

For which fragment of the logic does this proof search strategy work? An-
swering this question will give us (a slight generalization) of Horn clauses,
and those are the only ones allowed in backward chaining logic program-
ming languages like Prolog.

A key seems to be that the antecedents all have noninvertible left rules
that can be chained together. Similarly, we would like the succedent to
have noninvertible right rules, again so we can chain them (not visible in
this example). Remembering polarities: those propositions with invertible
right rules are negative and have noninvertible left rules, while those with
invertible left rules have noninvertible right rules are positive. We write D−

for the negative propositions in the antecedent1, and G+ for the positive
proposition in the succedent. We concentrate on the core propositions in
red and leave out some additional propositions, which the theory would
predict are compatible with backward chaining.

Program formulas D− ::= P− | G+ ⊃D− | ∀x.D−(x) | D−
1 ∧D−

2 | >
Programs Γ− ::= · | Γ−, D−

Goal formulas G+ ::= ↓P− | G+
1 ∧G+

2 | > | ∃x.G+(x) | G+
1 ∨G+

2 | ⊥

1D stands for “definitive clauses” from the early days of Prolog.

There may be a couple of surprises here. One is that conjunction is both
positive and negative. This is because there two forms of left rules for con-
junction, one that decomposes A∧B to A,B (which is invertible, and there-
fore positive) and one that extracts either A or B (which is noninvertible
and therefore negative). We have three judgment forms

Γ− f−→ P− stable sequent

Γ−, [D−]
f−→ P− left focus

Γ− f−→ [G+] right focus

It is also important to consider that we left out, which is (syntactically) very
little: program formulas D− cannot be of the form ↑G+, and goal formulas
G+ can neither be positive atoms P+ nor of the form ↓D− (except ↓P−).

Program formulas D− ::= · · · | ↑G−

Goal formulas G+ ::= · · · | P+ | ↓D−

By forcing goals to be almost entirely positive and programs to be entirely
negative, no inversion will ever be applied. The fact that we omitted pos-
itive atoms means chaining is always backwards (see next lecture). Essen-
tially, backward chaining logic programming arises from a pure backward
chaining interpretation of intuitionistic logic.

Another interesting aspect is that Horn clauses are so restricted that
classical and intuitionistic logic coincide on them. That is, a goal P− is
provable from a program Γ− in classical logic if and only if the sequent
is provable in intuitionistic logic. So at least in logic programming, there
shouldn’t be any arguments between intuitionists and classical logicians.

We now restate the rules for core Horn clauses, using our new notation.
Also, for existential goals, we introduce the ∃R rule. For both ∃R∗ and ∀L∗

we need a globally fresh variable X ; the superscript ∗ is supposed to be a
reminder of this condition.

D− ∈ Γ− Γ, [D−]
f−→ P−

Γ− f−→ P−
focusL

Γ−, [D−(X)]
f−→ P−

Γ−, [∀x.D−(x)]
f−→ P−

∀L∗
Γ−, [D−]

f−→ P− Γ− f−→ [G+]

Γ−, [G+ ⊃D−]
f−→ P−

⊃L

Q− = P−

Γ−, [Q−]
f−→ P−

id
no rule if Q− 6= P−

Γ−, [Q−]
f−→ P−

Γ− f−→ [G+
1 ] Γ− f−→ [G+

2 ]

Γ− f−→ [G+
1 ∧G+

2 ]

∧R
Γ− f−→ [>]

>R

Γ− f−→ [G(X)]

Γ− f−→ [∃x.G+(x)]
∃R∗ Γ− f−→ P−

Γ− f−→ [↓P−]
blur

6 Writing a Meta-Interpreter

Now that we have rewritten the foundation of logic programming in propo-
sitional form, can we take advantage of this for an implementation? An
ideal language for such an implementation would already support unifica-
tion and backtracking, since both of these are at the heart of the computation-
as-proof-search paradigm. We don’t have to look far: let’s use Prolog! This
means we are writing what is called a meta-circuluar interpreter: we provide
the semantics of a language in itself. There can be some arguments whether
this counts as “definitional” since, for example, the meta-interpreter will
not explain the details of unification because they are just inherited from
the metalanguage. Nevertheless, we can modify the semantics of chaining
by modifying the meta-interpreter, and we will play through an example
of this.

Now consider a program Γ− and a query P−, which should proceed by

searching for a proof of Γ− f−→ P−. In the remainder of this section we
will just omit the polarity annotations, remembering the program clauses
D and atoms P,Q are negative goals G are positive.

We begin with program representation, that is, the representation of Γ.
The idea is that every element D of Γ is represented in Prolog as a fact
prog(D). For example,

prog(inc(e,b1(e))).

This clause contains no variables. For the next one, ∀m. inc(b0(m), b1(m)),
we have to determine how to represent the quantifier. Unfortunately, this is
difficult in Prolog because it leaves quantifiers implicit and considers free
variables in the program to be universally quantified. We exploit this, by
instantiating m with a new free variable M and stating

prog(inc(b0(M),b1(M))).

For Horn clauses as we have defined them, it is always possible to move
the universal quantifier in program formulas D to the outside.

Finally, the last proposition in Γinc

∀m.∀n. inc(m,n)⊃ inc(b1(m), b0(n))

requires an implication. We write this in prefix form, G⊃D as imp(G,D),
so we don’t get confused between the language we are interpreting imp(G,D)
and the language in which we implement it in, whose implication would
be written as D :- G.

prog(imp(inc(M,N),inc(b1(M),b0(N)))).

Again, the quantified variables become free variables.
A priori, we could think the proof search would be through a predicate

solve(Gamma,G). But the program Γ never changes during search in the
Horn fragment, so it is sufficient to consider just solve(G). The first two
clauses should be clear: they implement ∧R and >R:

solve(true).
solve(and(G1,G2)) :- solve(G1), solve(G2).

The third possibility is an atom. We did not complicate the syntax with
an explicit atom constructor (perhaps we should have . . .), so we have a
predicate atm/1 that recognizes atoms. In our example:

atm(inc(_,_)).

After we have recognized that we are trying to prove an atom, we have to

nondeterministically select a program clause D and then see if Γ, [D]
f−→ P .

Selecting D takes place by calling prog(D), and the Γ, [D]
f−→ P will be

implemented by focus(D, P). Γ can remain implicit, because it is repre-
sented as facts in Prolog as explained before.

solve(P) :- atm(P), prog(D), focus(D, P).

The focus/2 predicate now distinguishes the cases for its first argument.
Since quantifiers are implicit, at this point this only pertains to implications
and atoms. For atoms, we unify them and make sure no clause applies if
they don’t unify.

focus(Q,P) :- atm(Q), Q = P.

For implications, we continue to focus on D first, and if that eventually
succeeds, we solve the subgoal.

focus(imp(G,D),P) :- focus(D,P), solve(G).

At this point, we already have a complete meta-interpreter for the Horn
clause fragment with prefixed quantifiers. Here is the summary.

solve(true).
solve(and(G1,G2)) :- solve(G1), solve(G2).
solve(P) :- atm(P), prog(D), focus(D, P).
focus(Q,P) :- atm(Q), Q = P.
focus(imp(G,D),P) :- focus(D,P), solve(G).

Our example of binary addition:

atm(inc(_,_)).

prog(inc(e,b1(e))).
prog(inc(b0(M),b1(M))).
prog(imp(inc(M,N),inc(b1(M),b0(N)))).

We can now exercise the meta-interpreter, again using free variables in
the query to model existentially quantified variables in the query.

| ?- solve(inc(b1(e),N)).

N = b0(b1(e)) ? ;

(1 ms) no
| ?- solve(inc(M,b1(e))).

M = e ? ;

M = b0(e) ? ;

no

With the second example we also see that Prolog backtracking implements
backtracking in our small Horn clause language.

7 Repairing Unsound Unification

One unfortunate consequence of using Prolog as our implementation lan-
guage is that we inherit its unsound unification. This means as a logic-
based proof search engine, our meta-interpreter has a severe bug. Fortu-
nately, Prolog also provides us with the means to fix it.

As an example, consider the query which should have no solution:
there is no m such that inc(b0(m), b1(b0(m)))! But instead of failing, it uses
the second clause and incorrectly succeeds in unifying M = b0(M), creat-
ing a circular term M = b0(b0(b0(. . .)))

?- solve(inc(b0(M),b1(b0(M)))).

cannot display cyclic term for M ? ;

no

If we want to make the program sound, we have to scour the meta-
interpreter for (a) explicit calls to unification, or (b) implicit unifications
which arise from repeated variables in the heads of clauses. We see there
are no repeated variables, and the only explicit call to unification is in the
case of atoms. So we can just replace this with a call to a library predicate
that implements sound unification.

solve(true).
solve(and(G1,G2)) :- solve(G1), solve(G2).
solve(P) :- atm(P), prog(D), focus(D, P).
% focus(Q,P) :- atm(Q), Q = P. % unsound unification
focus(Q,P) :- atm(Q), unify_with_occurs_check(Q,P).
focus(imp(G,D),P) :- focus(D,P), solve(G).

Now the previous query fails, as we had hoped.

?- solve(inc(b0(M),b1(b0(M)))).

no

8 Further Variants of the Semantics

We can now add the remaining connectives from our definitions of positive
and negative propositions, leaving only the quantifiers implicit. Note that
there are two clauses for disjunctive goals, and two clauses for conjunctive
programs. We could also have implemented this using a goal A ; B in
Prolog, but we prefer to keep the meta-language constructs simple.

Just to show what can be done with a metainterpreter, we also add a
minimal tracing facility. We do through via a predicate display/1 which
outputs its argument, and nl/0 which outputs a newline. Note that fail
is necessary so that Prolog backtracks after it has printed the goal and uses
the other clauses for solve/1 to actually solve the goal. Other variations
on this can be easily imagined.

% solve(G) succeeds if Gamma --> G
% focus(D,Q) succeeds if Gamma, [D] --> Q
solve(G) :- display(G), nl, fail.

solve(true).
solve(and(G1,G2)) :- solve(G1), solve(G2).
solve(or(G1,G2)) :- solve(G1).
solve(or(G1,G2)) :- solve(G2).
% no clause for solve(false)
solve(P) :- atm(P), prog(D), focus(D,P).

focus(imp(G,D),P) :- focus(D,P), solve(G).
focus(and(D1,D2),P) :- focus(D1, P).
focus(and(D1,D2),P) :- focus(D2, P).
focus(Q,P) :- atm(Q), unify_with_occurs_check(Q,P).

For example, the following query prints the three atomic goals it en-
counters in sequence (due to the carry bit), in each case with fresh internally
named existential variables.

?- solve(inc(b1(b1(e)),N)).
inc(b1(b1(e)),_283)
inc(b1(e),_302)
inc(e,_315)

N = b0(b0(b1(e))) ? ;

no

As a final possibility, consider how you might instrument the inter-
preter to not just present an answer substitution, but a proof term. The
solve/1 predicate would then be generalized to solve/2 and we would
ask, for example

solve(M,inc(e,N))

for some goal which would not only show N = b1(e) but also a proof
M : inc(e, b1(e)) that was found by the interpreter.

